

وقفه‌های سخت‌افزاری در میکروکنترلر ۸۰۵۱

مثال (۱) برنامه‌ای بنویسید که لامپ متصل به P1.0 را روشن و خاموش کند. اگر کلیدی به صورت زیر فشرده شود، باید لامپ به مدت دو برابر زمان عادی روشن بماند و بعد روشن و خاموش شدن آن ادامه یابد.
(۱) روش Polling: کلید را به P2.0 متصل می‌کنیم.

```
ORG 0
SETB P2.0 ; Define as input
```

```
AGAIN:
CPL P1.0
ACALL DELAY
```


```
JNB P2.0,ON
```

```
SJMP AGAIN
```

```
ON:
SETB P1.0
ACALL DELAY
ACALL DELAY
CLR P1.0
SJMP AGAIN
```

```
DELAY: MOV R0,#255
LOOP: NOP
DJNZ R0,LOOP
RET
```

```
END
```



مشکلات این سیستم:

- ۱- در هر دور اجرای برنامه باید فشرده شدن کلید چک شود (سربار اضافی).
- ۲- اگر در زمان اجرای DELAY کلید فشرده و رها شود، به آن اهمیتی داده نمی‌شود.

(۲) وقفه حساس به سطح: کلید را به یکی از دو پین وقفه مثلاً INTO متصل می‌کنیم.

```
ORG 0
LJMP MAIN
```

```
ORG 0003H
LJMP ISR_INT0
```

این خط می‌تواند حذف شود.

```
MAIN:
MOV IE,#10000001B ; Enable INTO
```

```
AGAIN:
CPL P1.0
ACALL DELAY
SJMP AGAIN
```

```
DELAY: MOV R0,#255
LOOP: NOP
DJNZ R0,LOOP
RET
```

```
ISR_INT0:
SETB P1.0
ACALL DELAY
ACALL DELAY
CLR P1.0
RETI
```

```
END
```

ویژگیها:

- بر خلاف حالت Polling، سربار اضافی وجود ندارد.
- در این سیستم تا وقتی دستمان روی کلید است، چراغ روشن می‌ماند.
- باید تا آغاز ISR دستمان را روی کلید نگه داریم؛ وگرنه به آن اهمیتی داده نمی‌شود.

۳) وقفه حساس به لبه : کلید را به یکی از دو پین وقفه مثلاً به INT1 متصل می‌کنیم.

```
ORG 0
LJMP MAIN
```


```
ORG 0013H
; ISR of INT1
SETB P1.0
ACALL DELAY
ACALL DELAY
CLR P1.0
RETI
```

```
MAIN:
MOV IE,#10000100B ; Enable INT1
SETB IT1 ; Make INT1 an Edge-Triggered interrupt.
```

```
AGAIN:
CPL P1.0
ACALL DELAY
SJMP AGAIN
```

```
DELAY: MOV R0,#255
LOOP: NOP
DJNZ R0,LOOP
RET
```

END



- بر خلاف حالت Polling، سربار اضافی وجود ندارد.
- در این سیستم اگر دستمان را روی کلید نگهداریم، تنها یک بار عمل روشن ماندن چراغ انجام می‌شود. برای تکرار عمل باید دستمان را از روی کلید برداریم و دو باره کلید را فشار دهیم (چون وقفه حساس به لبه است).
- تنها یک لحظه فشردن کلید کافی است و نیازی به نگه داشتن دست تا زمان آغاز ISR نمی‌باشد.

مثال ۲) مثالی از کاربرد همزمان دو وقفه سخت افزاری.

```
ORG 0
LJMP MAIN
```

```
ORG 0003H
LJMP ISR_INT0
```

```
ORG 0013H
LJMP ISR_INT1
```

```
MAIN:
MOV IE,#10000101B ; Enable INT1 & INT0
MOV R0,#200
```

```
AGAIN:
CPL P1.0
ACALL DELAY
SJMP AGAIN
```

```
ISR_INT0:
MOV R0,#100
RETI
```

```
ISR_INT1:
MOV R0,#200
RETI
```

```
DELAY:
MOV R1,R0
LOOP:
NOP
NOP
DJNZ R1,LOOP
RET
```

END